

graph twoway lpolyci — Local polynomial smooth plots with CIs

[Description](#)
[Options](#)

[Quick start](#)
[Remarks and examples](#)

[Menu](#)
[Also see](#)

[Syntax](#)

Description

`graph twoway lpolyci` plots a local polynomial smooth of *yvar* on *xvar* by using `graph twoway line` (see [G-2] [graph twoway line](#)), along with a confidence interval by using `graph twoway rarea` (see [G-2] [graph twoway rarea](#)).

Quick start

Local polynomial smooth plot of *y* versus *x* with 95% confidence interval and local mean smoothing

```
twoway lpolyci y x
```

As above, but with a polynomial of degree 3

```
twoway lpolyci y x, degree(3)
```

Overlay scatterplot of the observed data on polynomial smooth plot

```
twoway lpolyci y x || scatter y x
```

As above, but with a 99% confidence interval

```
twoway lpolyci y x, level(99) || scatter y x
```

Specify the Gaussian kernel function

```
twoway lpolyci y x, kernel(gaussian) || scatter y x
```

Specify the half-width of the kernel as 85

```
twoway lpolyci y x, bwidth(85) || scatter y x
```

With a separate graph area for each level of *catvar*

```
twoway lpolyci y x || scatter y x, by(catvar)
```

Add “My Title” as the title of the graph

```
twoway lpolyci y x || scatter y x, title("My Title")
```

Draw the confidence limits as a pair of lines rather than as an area

```
twoway lpolyci y x, ciplot(rline) || scatter y x
```

Menu

Graphics > Twoway graph (scatter, line, etc.)

Syntax

```
twoway lpolyci yvar xvar [if] [in] [weight] [, options]
```

<i>options</i>	Description
<code>kernel(<i>kernel</i>)</code>	kernel function; default is <code>kernel(epanechnikov)</code>
<code>bwidth(#)</code>	kernel bandwidth
<code>degree(#)</code>	degree of the polynomial smooth; default is <code>degree(0)</code>
<code>n(#)</code>	obtain the smooth at # points; default is <code>min(N, 50)</code>
<code>level(#)</code>	set confidence level; default is <code>level(95)</code>
<code>pwidth(#)</code>	pilot bandwidth for standard error calculation
<code>var(#)</code>	estimate of the constant conditional variance
<code>nofit</code>	do not plot the smooth
<code>fitplot(<i>plotype</i>)</code>	how to plot the smooth; default is <code>fitplot(line)</code>
<code>ciplot(<i>plotype</i>)</code>	how to plot CIs; default is <code>ciplot(rarea)</code>
<i>fcline_options</i>	change look of the smoothed line
<i>fitarea_options</i>	change look of CI
<i>axis_choice_options</i>	associate plot with alternative axis
<i>twoway_options</i>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

<i>kernel</i>	Description
<code>epanechnikov</code>	Epanechnikov kernel function; the default
<code>epan2</code>	alternative Epanechnikov kernel function
<code>biweight</code>	biweight kernel function
<code>cosine</code>	cosine trace kernel function
<code>gaussian</code>	Gaussian kernel function
<code>parzen</code>	Parzen kernel function
<code>rectangle</code>	rectangle kernel function
<code>triangle</code>	triangle kernel function

`fweights` and `awweights` are allowed; see [\[U\] 11.1.6 weight](#).

Options

`kernel(kernel)` specifies the kernel function for use in calculating the weighted local polynomial estimate. The default is `kernel(epanechnikov)`. See [\[R\] **kdensity**](#) for more information on this option.

`bwidth(#)` specifies the half-width of the kernel, the width of the smoothing window around each point. If `bwidth()` is not specified, a rule-of-thumb bandwidth estimator is calculated and used; see [\[R\] **lpoly**](#).

`degree(#)` specifies the degree of the polynomial to be used in the smoothing. The default is `degree(0)`, meaning local mean smoothing.

`n(#)` specifies the number of points at which the smooth is to be evaluated. The default is $\min(N, 50)$, where N is the number of observations.

`level(#)` specifies the confidence level, as a percentage, for confidence intervals. The default is `level(95)` or as set by `set level`; see [U] 20.7 Specifying the width of confidence intervals.

`pwidth(#)` specifies the pilot bandwidth to be used for standard error computations. The default is chosen to be 1.5 times the value of the rule-of-thumb bandwidth selector.

`var(#)` specifies an estimate of a constant conditional variance required for standard error computation. By default, the conditional variance at each smoothing point is estimated by the normalized weighted residual sum of squares obtained from locally fitting a polynomial of order $p + 2$, where p is the degree specified in `degree()`.

`nofit` prevents the smooth from being plotted.

`fitplot(plotype)` specifies how the prediction is to be plotted. The default is `fitplot(line)`, meaning that the smooth will be plotted by `graph twoway line`. See [G-2] graph twoway for a list of *plotype* choices. You may choose any that expects one y and one x variable. `fitplot()` is seldom used.

`ciplot(plotype)` specifies how the confidence interval is to be plotted. The default is `ciplot(rarea)`, meaning that the confidence bounds will be plotted by `graph twoway rarea`.

A reasonable alternative is `ciplot(rline)`, which will substitute lines around the smooth for shading. See [G-2] graph twoway for a list of *plotype* choices. You may choose any that expects two y variables and one x variable.

fcline_options specify how the `lpoly` line is rendered and its appearance; see [G-3] fcline_options.

fitarea_options specify how the confidence interval is rendered; see [G-3] fitarea_options. If you specify `ciplot()`, you should specify whatever is appropriate instead of using *fitarea_options*.

axis_choice_options associate the plot with a particular y or x axis on the graph; see [G-3] axis_choice_options.

twoway_options are a set of common options supported by all `twoway` graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [G-3] twoway_options.

Remarks and examples

[stata.com](http://www.stata.com)

`graph twoway lpolyci yvar xvar` uses the `lpoly` command—see [R] lpoly—to obtain a local polynomial smooth of *yvar* on *xvar* and confidence intervals and uses `graph twoway line` and `graph twoway rarea` to plot results.

Remarks are presented under the following headings:

Typical use


Use with by()

Typical use

`graph twoway lpolyci` can be used to overlay the confidence bands obtained from different local polynomial smooths. For example, for local mean and local cubic polynomial smooths:

```
. use http://www.stata-press.com/data/r14/auto
(1978 Automobile Data)

. twoway lpolyci weight length, nofit           ||
  lpolyci weight length, degree(3) nofit      ||
  ciplot(rline) pstyle(ci2) ||
  scatter weight length, msymbol(o)
```




The plotted area corresponds to the confidence bands for the local mean smooth and lines correspond to confidence intervals for the local cubic smooth.

When you overlay graphs, you nearly always need to respecify the axis titles by using the `axis_title_options` `ytitle()` and `xtitle()`; see [\[G-3\] axis_title_options](#).

Use with by()

graph twoway lpolyci may be used with by():

```
. use http://www.stata-press.com/data/r14/auto, clear
(1978 Automobile Data)
. twoway lpolyci weight length ||
    scatter weight length, msymbol(o) ||
    , by(foreign)
```



Also see

[R] [lpoly](#) — Kernel-weighted local polynomial smoothing

[G-2] [graph twoway lpolyci](#) — Local polynomial smooth plots with CIs